F01QKF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F01QKF returns the first *nrowp* rows of the real n by n orthogonal matrix P^T, where P is given as the product of Householder transformation matrices.

This routine is intended for use following F01QJF.

2 Specification

```
SUBROUTINE F01QKF(WHERET, M, N, NROWP, A, LDA, ZETA, WORK, IFAIL)
INTEGER           M, N, NROWP, LDA, IFAIL
real              A(LDA,*), ZETA(*), WORK(*)
CHARACTER*1       WHERET
```

3 Description

P is assumed to be given by

$$P = P_m P_{m-1} \dots P_1.$$

where

$$\mathrm{P}_k = I - u_k u_k^T,$$

$$\mathrm{u}_k = \begin{pmatrix} w_k \\ \zeta_k \\ 0 \\ z_k \end{pmatrix},$$

ζ_k is a scalar, w_k is a $(k-1)$ element vector and z_k is an $(n-m)$ element vector. w_k must be supplied in the kth row of A in elements A$(k,1)$,...,A$(k,k-1)$, z_k must be supplied in the kth row of A in elements A$(k,m+1)$,...,A(k,n) and ζ_k must be supplied either in A(k,k) or in ZETA(k), depending upon the parameter WHERET.

4 References

[1] Golub G H and van Loan C F (1996) *Matrix Computations* Johns Hopkins University Press (3rd Edition), Baltimore

[2] Wilkinson J H (1965) *The Algebraic Eigenvalue Problem* Oxford University Press, London

5 Parameters

1: WHERET — CHARACTER*1 *Input*

On entry: indicates where the elements of ζ are to be found as follows.

WHERET = 'I' (In A)

The elements of ζ are in A.

WHERET = 'S' (Separate)

The elements of ζ are separate from A, in ZETA.

Constraint: WHERET must be one of 'I' or 'S'.

2: M — INTEGER *Input*

On entry: m, the number of rows of A.

Constraint: M ≥ 0.

3: N — INTEGER *Input*

On entry: n, the number of columns of A.

Constraint: N $\geq$ M.

4: NROWP — INTEGER *Input*

On entry: the required number of rows of P, *nrowp*.

When NROWP = 0 then an immediate return is effected.

Constraint: $0 \leq$ NROWP $\leq$ N.

5: A(LDA,$*$) — ***real*** array *Input/Output*

Note: the second dimension of the array A must be at least max(1,N).

On entry: the leading m by m strictly lower triangular part of the array A, and the m by $(n-m)$ rectangular part of A with top left-hand corner at element A(1,M+1) must contain details of the matrix P. In addition, when WHERET = 'I', then the diagonal elements of A must contain the elements of ζ.

On exit: the first NROWP rows of the array A are overwritten by the first NROWP rows of the n by n orthogonal matrix P^T.

6: LDA — INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F01QKF is called.

Constraint: LDA $\geq$ max(1,M,NROWP).

7: ZETA($*$) — ***real*** array *Input*

Note. When WHERET = 'S', the dimension of the array ZETA must be at least max(1,M).

On entry: with WHERET = 'S', the array ZETA must contain the elements of ζ. If ZETA(k) = 0.0 then P_k is assumed to be I, otherwise ZETA(k) is assumed to contain ζ_k.

When WHERET = 'I', the array ZETA is not referenced.

8: WORK($*$) — ***real*** array *Workspace*

Note: the dimension of the array WORK must be at least max(M−1,NROWP−M,1).

9: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors detected by the routine:

IFAIL = −1

On entry, WHERET $\neq$ 'I' or 'S',
or M $<$ 0,
or N $<$ M,
or NROWP $<$ 0 or NROWP $>$ N,
or LDA $<$ max(M,NROWP).

7 Accuracy

The computed matrix P satisfies the relation

$$P = Q + E.$$

where Q is an exactly orthogonal matrix and

$$\|E\| \leq c\epsilon$$

where ϵ is the ***machine precision*** (see X02AJF), c is a modest function of n and $\|.\|$ denotes the spectral (two) norm. See also Section 7 of the document for F01QJF.

8 Further Comments

The approximate number of floating-point operations is given by

$$\frac{2}{3}m\{(3n-m)(2nrowp-m)-m(nrowp-m)\}, \quad nrowp \geq m$$

$$\frac{2}{3}nrowp^2(3n-nrowp), \qquad nrowp < m.$$

9 Example

To obtain the 5 by 5 orthogonal matrix P following the RQ factorization of the 3 by 5 matrix A given by

$$A = \begin{pmatrix} 2.0 & 2.0 & 1.6 & 2.0 & 1.2 \\ 2.5 & 2.5 & -0.4 & -0.5 & -0.3 \\ 2.5 & 2.5 & 2.8 & 0.5 & -2.9 \end{pmatrix}.$$

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F01QKF Example Program Text
*     Mark 14 Release.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NMAX, LDA, LDPT
      PARAMETER        (MMAX=10,NMAX=20,LDA=MMAX,LDPT=NMAX)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, J, M, N, NROWP
*     .. Local Arrays ..
      real             A(LDA,NMAX), PT(LDPT,NMAX), WORK(NMAX),
     +                 ZETA(NMAX)
*     .. External Subroutines ..
      EXTERNAL         F01QJF, F01QKF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F01QKF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      WRITE (NOUT,*)
      IF ((M.GT.MMAX) .OR. (N.GT.NMAX)) THEN
         WRITE (NOUT,*) 'M or N is out of range.'
         WRITE (NOUT,99999) 'M = ', M, '   N = ', N
      ELSE
```

```
         READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
         IFAIL = 0
*
*        Find the RQ factorization of A
         CALL F01QJF(M,N,A,LDA,ZETA,IFAIL)
*
*        Copy the array A into PT and form the n by n matrix conjg(P')
         DO 40 J = 1, N
            DO 20 I = 1, M
               PT(I,J) = A(I,J)
   20       CONTINUE
   40    CONTINUE
         NROWP = N
         IFAIL = 0
*
         CALL F01QKF('Separate',M,N,NROWP,PT,LDPT,ZETA,WORK,IFAIL)
*
         WRITE (NOUT,*) 'Matrix  P'
         DO 60 I = 1, N
            WRITE (NOUT,99998) (PT(J,I),J=1,NROWP)
   60    CONTINUE
      END IF
      STOP
*
99999 FORMAT (1X,A,I5,A,I5)
99998 FORMAT (5(1X,F8.4))
      END
```

9.2 Program Data

```
F01QKF Example Program Data
  3     5                        :Values of M and N
  2.0   2.0   1.6   2.0   1.2
  2.5   2.5  -0.4  -0.5  -0.3
  2.5   2.5   2.8   0.5  -2.9   :End of matrix A
```

9.3 Program Results

```
 F01QKF Example Program Results

 Matrix  P
  -0.1310  -0.5170  -0.4642  -0.5054  -0.4946
  -0.1310  -0.5170  -0.4642   0.5054   0.4946
  -0.3276   0.5499  -0.5199  -0.3957   0.4043
  -0.6551   0.2494  -0.0928   0.4946  -0.5054
  -0.6551  -0.3175   0.5385  -0.2967   0.3032
```